Thrust at NLO

Plot conditions

In[•]:=
```
Plot[{y2 /. Solve[y1 == 1 - y1 - y2, y2][[1]],
   y2 /. Solve[y2 == 1 - y1 - y2, y2][[1]], y1,
   y2 /. Solve[0 == 1 - y1 - y2, y2][[1]]}, {y1, 0, 1},
 AspectRatio → 1, Frame → True,
 PlotRange → {{0, 1}, {0, 1}}]
```

Out[•]=



Computation

To do the computation, replace conditions by theta-functions

In[∘]:=
```
prefact = CF (4 π)^(-1+ϵ) αs (μ2/Q2)^ϵ / Gamma[1 - ϵ];
```

In[∘]:=
```
rest = y1^(-1-ϵ) y2^(-1-ϵ) y3^(-ϵ)
    (2 (y1^2 + y2^2) (1 - ϵ) + 4 (y3 - y1 y2 ϵ));
```

In[∘]:=
```
rest0 = rest /. ϵ → 0
```

Out[∘]= $$\frac{2\left(y1^2+y2^2\right)+4\,y3}{y1\,y2}$$

Replace conditions by θ functions, perform δ-function integral:

In[∘]:=
```
y1smallest =
  rest0 UnitStep[y2 - y1] UnitStep[y3 - y1] /.
    y3 → 1 - y1 - y2 /. y1 → τ
```

Out[∘]= $$\frac{1}{y2\,\tau}\left(4\,(1-y2-\tau)+2\left(y2^2+\tau^2\right)\right)$$
UnitStep[1 - y2 - 2 τ] UnitStep[y2 - τ]

In[∘]:=
```
y2smallest =
  rest0 UnitStep[y1 - y2] UnitStep[y3 - y2] /.
    y3 → 1 - y1 - y2 /. y2 → τ
```

Out[∘]= $$\frac{1}{y1\,\tau}\left(4\,(1-y1-\tau)+2\left(y1^2+\tau^2\right)\right)$$
UnitStep[1 - y1 - 2 τ] UnitStep[y1 - τ]

In[∘]:=
```
y3smallest =
  rest0 UnitStep[y1 - y3] UnitStep[y2 - y3] /.
    y2 → 1 - y1 - y3 /. y3 → τ
```

Out[∘]= $$\frac{1}{y1\,(1-y1-\tau)}\left(2\left(y1^2+(1-y1-\tau)^2\right)+4\,\tau\right)$$
UnitStep[1 - y1 - 2 τ] UnitStep[y1 - τ]

In[∘]:=
```
y1smallest2 = Integrate[y1smallest, {y2, 0, 1},
   Assumptions → {τ > 0, τ < 1 / 3}]
```

Out[∘]=
$$\frac{-3+8\,\tau+3\,\tau^2+2\,\left(2-2\,\tau+\tau^2\right)\,\mathrm{Log}\left[-2+\frac{1}{\tau}\right]}{\tau}$$

In[∘]:=
```
y2smallest2 = Integrate[y2smallest, {y1, 0, 1},
   Assumptions → {τ > 0, τ < 1 / 3}]
```

Out[∘]=
$$\frac{-3+8\,\tau+3\,\tau^2+2\,\left(2-2\,\tau+\tau^2\right)\,\mathrm{Log}\left[-2+\frac{1}{\tau}\right]}{\tau}$$

In[∘]:=
```
y3smallest2 = Integrate[y3smallest, {y1, 0, 1},
   Assumptions → {τ > 0, τ < 1 / 3}]
```

Out[∘]=
$$-4+12\,\tau-\frac{4\,\left(1+\tau^2\right)\,\mathrm{Log}\left[-2+\frac{1}{\tau}\right]}{-1+\tau}$$

Now add up...

In[∘]:=
```
dSigmadτ =
  FullSimplify[y1smallest2 + y2smallest2 + y3smallest2,
   {τ > 0, τ < 1 / 3}]
```

Out[∘]=
$$\frac{2\,\left(3\,(1+\tau)\,(-1+3\,\tau)-\frac{2\,(2+3\,(-1+\tau)\,\tau)\,\mathrm{Log}\left[-2+\frac{1}{\tau}\right]}{-1+\tau}\right)}{\tau}$$

Note that $\tau < 1/3$ at NLO, otherwise the cross section is zero.

In[∘]:=
```
Series[dSigmadτ, {τ, 0, -1}]
```

Out[∘]=
$$\frac{2\,\left(-3+4\,\mathrm{Log}\left[\frac{1}{\tau}\right]\right)}{\tau}+O[\tau]^0$$